

ALU_UVM

Omar Magdy

AGENDA

Project Path

ALU Design

Environment Architecture

- Sequence Item
- Sequence
- Sequencer
- Driver
- Monitor
- Agent
- Scoreboard
- Coverage Collector
- Environment
- Test
- Top

Assertions

Simulation Results

Assertions Closure

Functional Coverage Closure

Code Coverage Closure

ALU

The verification environment is designed to fully verify ALU.

It has 10 ports, 9 input ports and 1 output port.

There are three operation sets, each of them are used depending on the different combinations of “a_en” and “b_en”.

```
module ALU #(parameter DATA_WIDTH = 5, OUTPUT_WIDTH = 6, A_OP_WIDTH = 3, B_OP_WIDTH = 2)(
    input clk,    // Clock
    input rst_n,  // Asynchronous reset active lo
    input ALU_en,   //System enable
    input a_en, b_en, //operations enable
    input [A_OP_WIDTH-1:0] a_op,
    input [B_OP_WIDTH-1:0] b_op,
    input signed [DATA_WIDTH-1:0] A, B,
    output logic [OUTPUT_WIDTH-1:0] C

);

    always_ff @(posedge clk or negedge rst_n) begin

        if (~rst_n) begin
          C <= 'd0;
        end
        else if (ALU_en) begin                          //ALU_1
            if (a_en && !b_en) begin
                case (a_op)
                    3'd0 : C <= {A[4], A} + {B[4], B};                  //ALU_2
                    3'd1 : C <= {A[4], A} - {B[4], B};                  //ALU_3
                    3'd2 : C <= A ^ B;                  //ALU_4
                    3'd3 : C <= A & B;                  //ALU_5
                    3'd4 : C <= A & B;                  //ALU_6
                    3'd5 : C <= A | B;                  //ALU_7
                    3'd6 : C <= ~(A ^ B);                   //ALU_8
                    default : C <= 'd0;                     //ALU_9
                endcase
            end

            else if (!a_en && b_en) begin
                case (b_op)
                    2'd0 : C <= ~(A & B);               //ALU_10
                    2'd1 : C <= {A[4], A} + {B[4], B};                  //ALU_11
                    2'd2 : C <= {A[4], A} + {B[4], B};                  //ALU_12
                    default : C <= 'd0;                     //ALU_13
                endcase
            end

            else if (a_en && b_en) begin
                case (b_op)
                    2'd0 : C <= A ^ B;                  //ALU_14
                    2'd1 : C <= ~(A ^ B);               //ALU_15
                    2'd2 : C <= {A[4], A} - 6'd1;               //ALU_16
                    2'd3 : C <= {B[4], B} + 6'd2;               //ALU_17
                endcase
            end
        end

    end

endmodule : ALU
```

CONFIGURATION OBJECT

Configuration object class is used to wrap variables that are going to be used later in various classes.

```
class ALU_cfg extends uvm_object;

    `uvm_object_utils(ALU_cfg)

    int item_count;
    int item_count_a_op;
    int item_count_b_op_1;
    int item_count_b_op_2;

    int item_count_sum = item_count_a_op + item_count_b_op_1 + item_count_b_op_2 + item_count;


    int count;
    int count_a_op;
    int count_b_op_1;
    int count_b_op_2;

    int count_sum;


    bit seq_a_op_on, seq_b_op_1_on, seq_b_op_2_on, seq_all_on;

    int correct_count, error_count;

    int cov_items;

    function new(string name = "ALU_cfg");
        super.new(name);
        item_count_a_op = 4;
        item_count_b_op_1 = 4;
        item_count_b_op_2 = 4;
        item_count = 4;

        count_a_op = 0;
        count_b_op_1 = 0;
        count_b_op_2 = 0;
        count = 0;

        count_sum = 0;

        seq_a_op_on = 0;
        seq_b_op_1_on = 0;
        seq_b_op_2_on = 0;
        seq_all_on = 0;

        correct_count = 0;
        error_count = 0;

        cov_items = 0;


    endfunction
endclass : ALU_cfg
```

PACKAGE

Package file is created to group common parameters that are used in multiple classes like input bit width, maximum positive, and negative values.

Also, it contains defined enums to facilitate dealing with the operations.

```
package p_headers;

    parameter DATA_WIDTH = 5, OUTPUT_WIDTH = 6, A_OP_WIDTH = 3, B_OP_WIDTH = 2, MAXPOSOP = 30, MAXNEGOP = -30, IGNORE = -16, IGNORE_OP = -32;


    // typedef enum {MAXPOS = (((2**DATA_WIDTH)/2)-1), ZERO = 0, MAXNEG = -(((2**DATA_WIDTH)/2)-1)} e_perm;
    typedef enum {MAXPOS = (((2**DATA_WIDTH)/2)-1), ZERO = 0, MAXNEG = -(((2**DATA_WIDTH)/2)-1)} e_perm;

    typedef enum {ADD_A, SUB_A, XOR_A, AND_A_1, AND_A_2, OR_A, XNOR_A, INVALID_A} e_a_op;

    typedef enum {NAND_B_1, ADD1_B_1, ADD2_B_1, INVALID_B_1} e_b_op_1;

    typedef enum {XOR_B_2, XNOR_B_2, SUBONE_B_2, ADDTWO_B_2} e_b_op_2;


endpackage : p_headers
```

ENVIRONMENT ARCHITECTURE

The UVM environment consists of:

- Sequence item: required to have the data to be randomized.
- Sequence: responsible for defining the scenario in which the sequence item randomization depends on.
- Sequencer: used to drive sequence item from sequence to driver.
- Driver: receives data from sequencer and drives the interface signals.
- Monitor: samples the DUT signals and convert them to transaction level.
- Agent: encapsulates driver, monitor, and sequencer.
- Scoreboard: checks the transaction coming from monitor.
- Coverage Collector: samples the incoming transaction to calculate functional coverage.
- Environment: encapsulates scoreboard, coverage collector.
- Test: it determine which sequence will be run.
- Top: it connects the DUT to the testbench.


Top
Test
ENV
Agent
Scoreboard
Coverage collector
Sequencer
Sequence
seq_item
Driver
Monitor
Interface
DUT

SEQUENCE ITEM

Sequence item contains the data that is going to be randomized.

Also, it has the constraints in which the randomization will use to achieve the highest functional coverage.

Field macro is used to determine which data is going to be printed.

```
class ALU_sequence_item extends uvm_sequence_item;


    logic ALU_en;

    rand logic a_en, b_en; //operations enable

    rand logic [A_OP_WIDTH-1:0] a_op;
    rand logic [B_OP_WIDTH-1:0] b_op;

    rand logic signed [DATA_WIDTH-1:0] A, B;

    logic [OUTPUT_WIDTH-1:0] C;


    ////////////////////////////////////////////////////////////////////////////////

    `uvm_object_utils(ALU_sequence_item)

    function new (string name = "ALU_sequence_item");
        super.new(name);
    endfunction


    ////////////////////////////////////////////////////////////////////////////////
```

SEQUENCE ITEM

"do_print" function is overridden to facilitate sequence item debugging.

```
virtual function void do_print(uvm_printer printer);
    super.do_print(printer);
    printer.print_field("a_en", a_en, $bits(a_en), UVM_BIN);
    printer.print_field("b_en", b_en, $bits(b_en), UVM_BIN);
   if(a_en && !b_en) begin
        enum_a_op = e_a_op'(a_op);
        printer.print_string("a_op", enum_a_op.name());
        if(enum_a_op == ADD_A || enum_a_op == SUB_A) begin
            printer.print_field("A", A, $bits(A), UVM_DEC);
            printer.print_field("B", B, $bits(B), UVM_DEC);
            printer.print_field("C", $signed(C), $bits(C), UVM_DEC);
        end
        else if(enum_a_op == XOR_A || enum_a_op == AND_A_1 || enum_a_op == AND_A_2 || enum_a_op == OR_A || enum_a_op == XNOR_A) begin
            printer.print_field("A", A, $bits(A), UVM_BIN);
            printer.print_field("B", B, $bits(B), UVM_BIN);
            printer.print_field("C", C, $bits(C), UVM_BIN);
        end
    end
    else if(!a_en && b_en) begin
        enum_b_op_1 = e_b_op_1'(b_op);
        printer.print_string("b_op", enum_b_op_1.name());
        if(enum_b_op_1 == ADD1_B_1 || enum_b_op_1 == ADD2_B_1) begin
            printer.print_field("A", A, $bits(A), UVM_DEC);
            printer.print_field("B", B, $bits(B), UVM_DEC);
            printer.print_field("C", $signed(C), $bits(C), UVM_DEC);
        end
        else if(enum_b_op_1 == NAND_B_1) begin
            printer.print_field("A", A, $bits(A), UVM_BIN);
            printer.print_field("B", B, $bits(B), UVM_BIN);
            printer.print_field("C", C, $bits(C), UVM_BIN);
        end
    end
    else if (a_en && b_en) begin
        enum_b_op_2 = e_b_op_2'(b_op);
        printer.print_string("b_op", enum_b_op_2.name());
        if(b_op == SUBONE_B_2 || b_op == ADDTWO_B_2) begin
            printer.print_field("A", A, $bits(A), UVM_DEC);
            printer.print_field("B", B, $bits(B), UVM_DEC);
            printer.print_field("C", $signed(C), $bits(C), UVM_DEC);
        end
        else if(b_op == XOR_B_2 || b_op == XNOR_B_2) begin
            printer.print_field("A", A, $bits(A), UVM_BIN);
            printer.print_field("B", B, $bits(B), UVM_BIN);
            printer.print_field("C", C, $bits(C), UVM_BIN);
        end
    end
    else begin
        printer.print_field("A", A, $bits(A), UVM_HEX);
        printer.print_field("B", B, $bits(B), UVM_HEX);
        printer.print_field("C", C, $bits(C), UVM_HEX);
    end
    endfunction
```

SEQUENCE

The sequence determine the scenario for testing.

The sequence class has 4 scenarios:

- One for a_op set.
- One for the first b_op set.
- One for the second b_op set.
- One is a randomization sequence.

Final sequence is used to wrap and all sequences using uvm_do macro.

“on” signals are used from “cfg” to ensure the packets of specific sequences are counted in scoreboard class.

SEQUENCE

Sequence a_op

- The sequence turn off randomization for a_en and b_en and sets them to 1 and 0 respectively to enable the a_op set.
- It get the item count from configuration object set in the test.

```
class ALU_sequence_a_op extends uvm_sequence #(ALU_sequence_item);

    `uvm_object_utils(ALU_sequence_a_op)

    ALU_cfg cfg;

    function new(string name = "ALU_sequence_a_op");
        super.new(name);
    endfunction


    virtual task body ();
        if (!uvm_config_db#(ALU_cfg)::get(null, "", "ALU_cfg", cfg)) begin
          `uvm_fatal("CFG_ERR", "Failed to retrieve configuration object!")
        end
        cfg.seq_a_op_on = 1;
        cfg.seq_b_op_1_on = 0;
        cfg.seq_b_op_2_on = 0;
        cfg.seq_all_on = 0;


        repeat (cfg.item_count_a_op) begin
            req = ALU_sequence_item::type_id::create("req");
            req.a_en.rand_mode(0);
            req.b_en.rand_mode(0);

            start_item(req);
            req.a_en = 1;
            req.b_en = 0;
            assert(req.randomize());


            finish_item(req);
        end
        `uvm_info(get_full_name(), $sformatf("Done generation of %0d items", cfg.item_count_a_op), UVM_LOW)

    endtask

endclass : ALU_sequence_a_op
```

SEQUENCE

Sequence b_op_1

- The sequence turn off randomization for a_en and b_en and sets them to 0 and 1 respectively to enable the b_op set.
- It get the item count from configuration object set in the test.

```
class ALU_sequence_b_op_1 extends uvm_sequence #(ALU_sequence_item);

    `uvm_object_utils(ALU_sequence_b_op_1)

    ALU_cfg cfg;

    function new(string name = "ALU_sequence_b_op_1");
        super.new(name);
    endfunction


    virtual task body ();
        if (!uvm_config_db#(ALU_cfg)::get(null, "", "ALU_cfg", cfg)) begin
          `uvm_fatal("CFG_ERR", "Failed to retrieve configuration object!")
        end
        cfg.seq_a_op_on = 0;
        cfg.seq_b_op_1_on = 1;
        cfg.seq_b_op_2_on = 0;
        cfg.seq_all_on = 0;


        repeat (cfg.item_count_b_op_1) begin
            req = ALU_sequence_item::type_id::create("req");
            req.a_en.rand_mode(0);
            req.b_en.rand_mode(0);

            start_item(req);
            req.a_en = 0;
            req.b_en = 1;
            assert(req.randomize());


                finish_item(req);

        end
        `uvm_info(get_full_name(), $sformatf("Done generation of %0d items", cfg.item_count_b_op_1), UVM_LOW)

    endtask

endclass : ALU_sequence_b_op_1
```

SEQUENCE

Sequence b_op_2

- The sequence turn off randomization for a_en and b_en and sets them to 1 and 1 respectively to enable the b_op set.
- It get the item count from configuration object set in the test.

```
class ALU_sequence_b_op_2 extends uvm_sequence #(ALU_sequence_item);

    `uvm_object_utils(ALU_sequence_b_op_2)

    ALU_cfg cfg;

    function new(string name = "ALU_sequence_b_op_2");
        super.new(name);
    endfunction


    virtual task body ();
        if (!uvm_config_db#(ALU_cfg)::get(null, "", "ALU_cfg", cfg)) begin
          `uvm_fatal("CFG_ERR", "Failed to retrieve configuration object!")
        end
        cfg.seq_a_op_on = 0;
        cfg.seq_b_op_1_on = 0;
        cfg.seq_b_op_2_on = 1;
        cfg.seq_all_on = 0;
        repeat (cfg.item_count_b_op_2) begin
            req = ALU_sequence_item::type_id::create("req");
            req.a_en.rand_mode(0);
            req.b_en.rand_mode(0);

            start_item(req);
            req.a_en = 1;
            req.b_en = 1;
            assert(req.randomize());


            finish_item(req);

        end
        `uvm_info(get_full_name(), $sformatf("Done generation of %0d items", cfg.item_count_b_op_2), UVM_LOW)

    endtask

endclass : ALU_sequence_b_op_2
```

SEQUENCE

Sequence all

- The sequence randomizes “a_en” and “b_en”.

```
class ALU_sequence_all extends uvm_sequence #(ALU_sequence_item);

    `uvm_object_utils(ALU_sequence_all)

    ALU_cfg cfg;

    function new(string name = "ALU_sequence_all");
        super.new(name);
    endfunction


    virtual task body ();
        if (!uvm_config_db#(ALU_cfg)::get(null, "", "ALU_cfg", cfg)) begin
          `uvm_fatal("CFG_ERR", "Failed to retrieve configuration object!")
        end
        cfg.seq_a_op_on = 0;
        cfg.seq_b_op_1_on = 0;
        cfg.seq_b_op_2_on = 0;
        cfg.seq_all_on = 1;
        repeat (cfg.item_count) begin

            req = ALU_sequence_item::type_id::create("req");
            start_item(req);
            `uvm_info(get_full_name(), $sformatf("start_item: "), UVM_LOW)

            assert(req.randomize());
            `uvm_info(get_full_name(), $sformatf("Generate new item: "), UVM_LOW)

            finish_item(req);
        end
        `uvm_info(get_full_name(), $sformatf("Done generation of %0d items", cfg.item_count), UVM_LOW)

    endtask

endclass : ALU_sequence_all
```

SEQUENCE

Sequence wrapper

- The sequence initializes all sequences and run them sequentially.

```
class ALU_sequence extends uvm_sequence #(ALU_sequence_item);

    `uvm_object_utils(ALU_sequence)

    ALU_sequence_a_op seq_a_op;
    ALU_sequence_b_op_1 seq_b_op_1;
    ALU_sequence_b_op_2 seq_b_op_2;
    ALU_sequence_all seq_all;

    ALU_cfg cfg;


    function new(string name = "ALU_sequence");
        super.new(name);
    endfunction


    virtual task body ();
        if (!uvm_config_db#(ALU_cfg)::get(null, "", "ALU_cfg", cfg))
          `uvm_fatal("CFG_ERR", "Failed to retrieve configuration object!")

        `uvm_do(seq_a_op)
        `uvm_do(seq_b_op_1)
        `uvm_do(seq_b_op_2)
        `uvm_do(seq_all)

        `uvm_info(get_full_name(), $sformatf("Done generation of %0d items", cfg.item_count_sum), UVM_LOW)

    endtask

endclass : ALU_sequence
```

DRIVER

The driver transforms from transaction level to pin level.

```
function void build_phase(uvm_phase phase);
    super.build_phase(phase);
    if (!uvm_config_db#(virtual interface alu_if)::get(this, "", "my_vif", intf))
        `uvm_fatal(get_full_name(), "ERROR FETCHING my_vif")
    req_clone = ALU_sequence_item::type_id::create("req_clone");
endfunction


task run_phase(uvm_phase phase);
    super.run_phase(phase);
    forever begin
        seq_item_port.get_next_item(req);
        drive();
        seq_item_port.item_done();
    end
endtask


virtual task drive;
    req_clone = req;
    @(posedge intf.clk);
    `uvm_info(get_full_name(), $sformatf("Sample Inputs"), UVM_LOW)
    intf.ALU_en <= 1;
    intf.a_en <= req.a_en;
    intf.b_en <= req.b_en;
    intf.a_op <= req.a_op;
    intf.b_op <= req.b_op;
    intf.A <= req.A;
    intf.B <= req.B;

    @(posedge intf.clk);
    intf.ALU_en <= 0;
    req.C = intf.C;
    @(posedge intf.clk);

endtask
```

MONITOR

The Monitor is responsible for sampling interface values and assigning them to sequence item which will be sent to Scoreboard and Coverage collector classes.

```
function void build_phase(uvm_phase phase);
    super.build_phase(phase);
    if (!uvm_config_db#(virtual alu_if)::get(this, "", "my_vif", intf)) begin
        `uvm_fatal(get_full_name(), "ERROR FETCHING my_vif")
    end
endfunction


virtual task run_phase(uvm_phase phase);
    super.run_phase(phase);
    forever begin
        req = ALU_sequence_item::type_id::create("ALU_sequence_item");
        @(posedge intf.clk iff intf.ALU_en);
        req.ALU_en = intf.ALU_en;
        req.a_en = intf.a_en;
        req.b_en = intf.b_en;
        req.a_op = intf.a_op;
        req.b_op = intf.b_op;
        req.A = intf.A;
        req.B = intf.B;

        @(posedge intf.clk);
        req.C = intf.C;
        m_analysis_port.write(req);
    end
endtask : run_phase
```

AGENT

It encapsulates driver, monitor, and sequencer.

```
function void build_phase(uvm_phase phase);
    super.build_phase(phase);
    sqr = ALU_sequencer::type_id::create("sqr", this);
    driv = ALU_driver::type_id::create("driv", this);
    mon = ALU_monitor::type_id::create("mon", this);

    if (!uvm_config_db#(virtual interface alu_if)::get(this, "", "my_vif", intf)) begin
        `uvm_fatal(get_full_name(), "ERROR FETCHING my_vif")
    end

    uvm_config_db#(virtual interface alu_if)::set(this, "*", "my_vif", intf);


endfunction

function void connect_phase(uvm_phase phase);
    super.connect_phase(phase);
    driv.seq_item_port.connect(sqr.seq_item_export);
endfunction
```

SCOREBOARD

Scoreboard receives the transactions from Monitor class.

It samples transaction values to evaluate the expected output and check the actual output value sampled from the same transaction.

SCOREBOARD

```
virtual task run_phase(uvm_phase phase);
    super.run_phase(phase);

    forever begin

        ALU_sequence_item req_clone;

        wait(seq_item_q.size() > 0);
        req_clone = seq_item_q.pop_front();

        if (req_clone.a_en && !req_clone.b_en) begin
            case (req_clone.a_op)
                'd0 : C_expected = {req_clone.A[4], req_clone.A} + {req_clone.B[4], req_clone.B};              //ALU_2
                'd1 : C_expected = {req_clone.A[4], req_clone.A} - {req_clone.B[4], req_clone.B};              //ALU_3
                'd2 : C_expected = req_clone.A ^ req_clone.B;                                                 //ALU_4
                'd3 : C_expected = req_clone.A & req_clone.B;                                                 //ALU_5
                'd4 : C_expected = req_clone.A & req_clone.B;                                                 //ALU_6
                'd5 : C_expected = req_clone.A | req_clone.B;                                                 //ALU_7
                'd6 : C_expected = ~(req_clone.A ^ req_clone.B);                                              //ALU_8
                default : C_expected = 0;
            endcase
        end

        else if (!req_clone.a_en && req_clone.b_en) begin
            case (req_clone.b_op)
                'd0 : C_expected = ~(req_clone.A & req_clone.B);                                              //ALU_10
                'd1 : C_expected = {req_clone.A[4], req_clone.A} + {req_clone.B[4], req_clone.B};              //ALU_11
                'd2 : C_expected = {req_clone.A[4], req_clone.A} + {req_clone.B[4], req_clone.B};              //ALU_12
                default : C_expected = 0;
            endcase
        end

        else if (req_clone.a_en && req_clone.b_en) begin
            case (req_clone.b_op)
                'd0 : C_expected = req_clone.A ^ req_clone.B;                                                 //ALU_14
                'd1 : C_expected = ~(req_clone.A ^ req_clone.B);                                              //ALU_15
                'd2 : C_expected = {req_clone.A[4], req_clone.A} - 6'd1;                                      //ALU_16
                'd3 : C_expected = {req_clone.B[4], req_clone.B} + 6'd2;                                      //ALU_17
            endcase
        end
        else begin
            C_expected = C_expected;                                                                          //ALU_18
        end
```

```
        if (req_clone.C != C_expected) begin
          `uvm_info(get_type_name(),$sformatf("------ :: FAIL::DATA MISMATCHED! :: ------"),UVM_LOW)
          `uvm_info(get_type_name(),$sformatf("Expected Data: %0d Actual Data: %0d",$signed(C_expected), $signed(req_clone.C)),UVM_LOW)
          req_clone.print();
          `uvm_info(get_type_name(),"-----------------------------------",UVM_LOW)
          cfg.error_count++;
        end

        else begin
          `uvm_info(get_type_name(),$sformatf("------ :: SUCCESS::DATA MATCHED! :: ------"),UVM_LOW)
          `uvm_info(get_type_name(),$sformatf("Expected Data: %0d Actual Data: %0d",$signed(C_expected), $signed(req_clone.C)),UVM_LOW)
          req_clone.print();
          `uvm_info(get_type_name(),"-----------------------------------",UVM_LOW)
          cfg.correct_count++;
        end
      end
    endtask : run_phase
```

SCOREBOARD

The scoreboard increments the number of incoming packets from each sequence to verify all packets are received by the scoreboard.

```
virtual function void count_packets();
  if (cfg.seq_a_op_on) begin
    cfg.count_a_op++;
  end

  if (cfg.seq_b_op_1_on) begin
    cfg.count_b_op_1++;
  end

  if (cfg.seq_b_op_2_on) begin
    cfg.count_b_op_2++;
  end

  if (cfg.seq_all_on) begin
    cfg.count++;
  end


endfunction : count_packets
```

COVERAGE COLLECTOR

The class is responsible for grouping the cover points and cross coverage for the transaction values.

This ensures ALU functionality is fully checked.

The cover group is sampled when collector class receives the transaction.

Extract phase is written to print the final coverage percentage.

```
  function void build_phase(uvm_phase phase);
      super.build_phase(phase);
      if (~uvm_config_db#(ALU_cfg)::get(null, "", "ALU_cfg", cfg)) begin
          `uvm_fatal(get_full_name(), "ERROR FETCHING ALU_cfg")
      end
      seq_item = ALU_sequence_item::type_id::create("seq_item");
  endfunction



  function void write(ALU_sequence_item t);
      seq_item = t;
      cfg.cov_items++;
      ALU_cg.sample();
  endfunction

  virtual function void extract_phase(uvm_phase phase);
      super.extract_phase(phase);

      `uvm_info(get_type_name(), $sformatf("Coverage Report: %0f", ALU_cg.get_coverage()), UVM_LOW)
endfunction
```

COVERAGE COLLECTOR

Illegal bins are defined for the prohibited operations in set A and set B1.

Furthermore, illegal bins are defined for “-32” for “C” and “-16” for both “A” and “B”.

Ignore bins are defined for “-31,31” as they can not be achieved given the range of the inputs [-15, 15].

```
a_op_cp     : coverpoint seq_item.a_op iff (seq_item.a_en && !seq_item.b_en)
                                {
                                    bins add = {ADD_A};
                                    bins sub = {SUB_A};
                                    bins logic_process[] = {XOR_A, AND_A_1, AND_A_2, OR_A, XNOR_A};
                                    illegal_bins a_op_invalid = {INVALID_A};
                                }
```

```
                                }
b_op_1_cp       : coverpoint seq_item.b_op iff (!seq_item.a_en && seq_item.b_en)
                                {
                                    bins logic_process = {NAND_B_1};
                                    bins add[] = {ADD1_B_1, ADD2_B_1};
                                    illegal_bins b_op_invalid = {INVALID_B_1};
                                }
```

```
C_cp            : coverpoint $signed(seq_item.C)
                                {
                                    ignore_bins ignore = {-31, 31};
                                    illegal_bins illegal = {-32};
                                }
```

ENVIRONMENT

Environment class encapsulates Agent, Scoreboard, and Coverage Collector classes.

```
function void build_phase(uvm_phase phase);
    super.build_phase(phase);
    agt = ALU_agent::type_id::create("agt", this);
    scb = ALU_scoreboard::type_id::create("scb", this);
    cov = ALU_coverage_collector::type_id::create("cov", this);

    if (!uvm_config_db#(virtual interface alu_if)::get(this, "", "my_vif", intf)) begin
        `uvm_fatal(get_full_name(), "ERROR FETCHING my_vif!")
    end
    uvm_config_db#(virtual interface alu_if)::set(this, "agt", "my_vif", intf);



endfunction

function void connect_phase(uvm_phase phase);
    super.connect_phase(phase);


    agt.mon.m_analysis_port.connect(scb.m_analysis_imp);
    agt.mon.m_analysis_port.connect(cov.analysis_export);
endfunction
```

TEST

Test program creates the Environment class and starts the intended sequence.

The number of packets in each sequence are determined in build phase.

```
function void build_phase(uvm_phase phase);
    super.build_phase(phase);
    env = ALU_env::type_id::create("env", this);
    cfg = ALU_cfg::type_id::create("cfg");

    cfg.item_count = 1000;
    cfg.item_count_a_op = 2000;
    cfg.item_count_b_op_1 = 1000;
    cfg.item_count_b_op_2 = 1000;

    cfg.item_count_sum =  cfg.item_count_a_op + cfg.item_count_b_op_1 + cfg.item_count_b_op_2 + cfg.item_count;


    if (!uvm_config_db#(virtual interface alu_if)::get(this, "", "my_vif", intf)) begin
        `uvm_fatal(get_full_name(), "ERROR FETCHING my_vif!")
    end

    uvm_config_db #(virtual interface alu_if)::set(this, "env", "my_vif", intf);
    uvm_config_db #(ALU_cfg)::set(null, "*", "ALU_cfg", cfg);

    seq = ALU_sequence::type_id::create("seq");

endfunction


task run_phase(uvm_phase phase);
    super.run_phase(phase);
    phase.raise_objection(this);
    seq.start(env.agt.sqr);
    cfg.count_sum =  cfg.count_a_op + cfg.count_b_op_1 + cfg.count_b_op_2 + cfg.count;
    phase.drop_objection(this);

endtask
```

TEST

Phase ready to end function is used to ensure all packets are processed.

```
bit done;
function void phase_ready_to_end(uvm_phase phase);
    if (phase.is(uvm_run_phase::get)) begin
        if (done != 1) begin
            phase.raise_objection(this, "Test not ready yet");
            fork
                `uvm_info("PRTESTING", "Phase Ready Testing", UVM_LOW);
                wait_for_ready_to_finish(phase);
            join_none
        end
    end
endfunction : phase_ready_to_end

task wait_for_ready_to_finish(uvm_phase phase);

    `uvm_info(get_full_name(), $sformatf("cfg.cov_items = %0d", cfg.cov_items), UVM_LOW);
    wait(cfg.item_count_sum == cfg.count_sum);
    wait(cfg.item_count_sum == cfg.cov_items);

    done = 1;

    `uvm_info(get_full_name(), $sformatf("cfg.item_count_sum = %0d", cfg.item_count_sum), UVM_LOW);
    `uvm_info(get_full_name(), $sformatf("cfg.count_sum = %0d", cfg.count_sum), UVM_LOW);

    `uvm_info(get_full_name(), $sformatf("cfg.correct_count = %0d, cfg.error_count = %0d", cfg.correct_count, cfg.error_count), UVM_LOW);

    phase.drop_objection(this, "Test ready to end");


endtask: wait_for_ready_to_finish
```

TOP

Top module initiates DUT, Test, and interface.

Reset is applied before running the test.

System clock is generated and passed to the interface as an input.

Bind is used to initiate ALU_SVA module in ALU module.

```
module ALU_top ();

    import uvm_pkg::*;
    `include "uvm_macros.svh"

    import pack::*;

    bit clk, rst_n;

    alu_if intf(clk, rst_n);

    ALU DUT (
    .clk (intf.clk),
    .rst_n (intf.rst_n),
    .ALU_en(intf.ALU_en),
    .a_en  (intf.a_en),
    .b_en  (intf.b_en),
    .a_op  (intf.a_op),
    .b_op  (intf.b_op),
    .A     (intf.A),
    .B     (intf.B),
    .C     (intf.C)
   );

    always #5 clk = ~clk;

    initial begin
        rst_n = 0;
        #5 rst_n =1;
        #5 rst_n = 0;
        #5 rst_n =1;
    end

    bind ALU ALU_SVA ALU_SVA_inst0(intf);

    initial begin
        uvm_config_db#(virtual interface alu_if)::set(null, "uvm_test_top", "my_vif", intf);
    end

    initial begin
        run_test("ALU_test");
    end


endmodule : ALU_top
```

ASSERTIONS

Assertions module is created to group multiple properties of the design to fully ensure the properties of the design are covered.

```
s_ALU_2  : assert property(p_ALU_2);
s_ALU_3  : assert property(p_ALU_3);
s_ALU_4  : assert property(p_ALU_4);
s_ALU_5  : assert property(p_ALU_5);
s_ALU_6  : assert property(p_ALU_6);
s_ALU_7  : assert property(p_ALU_7);
s_ALU_8  : assert property(p_ALU_8);
s_ALU_10 : assert property(p_ALU_10);
s_ALU_11 : assert property(p_ALU_11);
s_ALU_12 : assert property(p_ALU_12);
s_ALU_14 : assert property(p_ALU_14);
s_ALU_15 : assert property(p_ALU_15);
s_ALU_16 : assert property(p_ALU_16);
s_ALU_17 : assert property(p_ALU_17);


c_ALU_2  : cover property(p_ALU_2);
c_ALU_3  : cover property(p_ALU_3);
c_ALU_4  : cover property(p_ALU_4);
c_ALU_5  : cover property(p_ALU_5);
c_ALU_6  : cover property(p_ALU_6);
c_ALU_7  : cover property(p_ALU_7);
c_ALU_8  : cover property(p_ALU_8);
c_ALU_10 : cover property(p_ALU_10);
c_ALU_11 : cover property(p_ALU_11);
c_ALU_12 : cover property(p_ALU_12);
c_ALU_14 : cover property(p_ALU_14);
c_ALU_15 : cover property(p_ALU_15);
c_ALU_16 : cover property(p_ALU_16);
c_ALU_17 : cover property(p_ALU_17);
```

SIMULATION RESULTS

The results is checked against reference logic implemented in scoreboard.

In addition, SVA is used to ensure the functionality of the DUT is error free.

SIMULATION RESULTS



SIMULATION RESULTS



SIMULATION RESULTS

5000 transactions are created in all sequences with all passing the given tests.

```
UVM_INFO ALU_test.svh(59) @ 149995000: uvm_test_top [PRTESTING] Phase Ready Testing
UVM_INFO ALU_test.svh(69) @ 149995000: uvm_test_top [uvm_test_top] cfg.cov_items = 5000
UVM_INFO ALU_test.svh(75) @ 149995000: uvm_test_top [uvm_test_top] cfg.item_count_sum = 5000
UVM_INFO ALU_test.svh(76) @ 149995000: uvm_test_top [uvm_test_top] cfg.count_sum = 5000
UVM_INFO ALU_test.svh(78) @ 149995000: uvm_test_top [uvm_test_top] cfg.correct_count = 5000, cfg.error_count = 0
UVM_INFO /tools/Synopsys/install/vcs/V-2023.12-1/etc/uvm/base/uvm_objection.svh(1274) @ 149995000: reporter [TEST_DONE] 'run' phase is ready to proceed to the 'extract'
phase
UVM_INFO ALU_coverage_collector.svh(157) @ 149995000: uvm_test_top.env.cov [ALU_coverage_collector] Coverage Report: 100.000000
```

ASSERTIONS CLOSURE

Using Assertions, all properties have passed with no errors when asserting the defined properties in the design.

COVER PROPERTIES	CATEGORY	SEVERITY	ATTEMPTS	MATCHES
ALU_top.DUT.ALU_SVA_inst0.c_ALU_10	0	0	14999	1416
ALU_top.DUT.ALU_SVA_inst0.c_ALU_11	0	0	14999	1398
ALU_top.DUT.ALU_SVA_inst0.c_ALU_12	0	0	14999	1539
ALU_top.DUT.ALU_SVA_inst0.c_ALU_14	0	0	14999	825
ALU_top.DUT.ALU_SVA_inst0.c_ALU_15	0	0	14999	921
ALU_top.DUT.ALU_SVA_inst0.c_ALU_16	0	0	14999	954
ALU_top.DUT.ALU_SVA_inst0.c_ALU_17	0	0	14999	1008
ALU_top.DUT.ALU_SVA_inst0.c_ALU_2	0	0	14999	975
ALU_top.DUT.ALU_SVA_inst0.c_ALU_3	0	0	14999	954
ALU_top.DUT.ALU_SVA_inst0.c_ALU_4	0	0	14999	993
ALU_top.DUT.ALU_SVA_inst0.c_ALU_5	0	0	14999	912
ALU_top.DUT.ALU_SVA_inst0.c_ALU_6	0	0	14999	918
ALU_top.DUT.ALU_SVA_inst0.c_ALU_7	0	0	14999	1029
ALU_top.DUT.ALU_SVA_inst0.c_ALU_8	0	0	14999	885

FUNCTIONAL COVERAGE CLOSURE

Regarding functional coverage, 100% achieved where multiple constraints are designed to achieve all possible combinations of inputs.

Total Groups Coverage Summary

SCORE	WEIGHT
100.00	1

Total groups in report: 1

NAME	SCORE	WEIGHT	GOAL	AT LEAST	PER INSTANCE	AUTO BIN MAX	PRINT MISSING	COMMENT
pack::ALU_coverage_collector::ALU_cg	100.00	1	100	1	0	64	64	

CODE COVERAGE CLOSURE

For code coverage, 100% is achieved.

However, multiple exclusions are added due to design requirements that resulted in multiple operations are not permitted to be used.

Also, missing default in third case statements is excluded as all cases are implemented and covered.

In addition, missing “else” in always block is excluded as sequential always block is used where no else statement is needed.

Total Coverage Summary

SCORE	LINE	COND	TOGGLE	BRANCH	ASSERT	GROUP
100.00	100.00	100.00	100.00	100.00	100.00	100.00



THANK YOU